準天頂衛星システム
quasi-zenith satellite system; QZSS

H. FUKUDA

College of Liberal Arts and Sciences, Kitasato University
1-15-1 Kitasato, Minamiku, Sagamihara, Kanagawa 252-0373, JAPAN

(2010.9.28)

回転行列の定義

```
Rz[t_] := {{Cos[t], Sin[t], 0}, {-Sin[t], Cos[t], 0}, {0, 0, 1}};
Rx[t_] := {{1, 0, 0}, {0, Cos[t], Sin[t]}, {0, -Sin[t], Cos[t]}};
```

地球の自転周期は2πとする.
原点を地球の重心，Z軸を自転軸（北極方向）にとった空間固定系をXYZとする.
Zを軸にX→Yへt回転した座標をxyzとする．これは地球固定系．tは時刻．変換行列Rz(t).

Zを軸にX→YへΦ回転した座標をX'Y'Zとする．変換行列Rz(Φ).
X'を軸にY'→ZへΘ回転した座標をX'Y"Z'とする．変換行列Rx(Θ).
Z'を軸にした，原点のまわりの周期2π，長径1の楕円運動が衛星の運動．Θは楕円運動の軸と地球の自転軸の角度.

```
Clear[Θ, Φ];
```

したがって，衛星の運動はX'Y"Z'座標でのX'Y"面上の運動.
Φ=0のときの運動をq(t)=(qx(t), qy(t), 0)とする．すると，どんなΦに対してもq(t-Φ)は地球固定系の同じ軌道にのる（藤原さんより）．楕円軌道のノーテーションは，ランダウ-リフシッツの力学と同じ.

```
Clear[e];
sp = Sqrt[1 - e^2];
ti[ξ_] := ξ - e Sin[ξ + Pi];
dtpdξ[ξ_] := 1 - e Cos[ξ + Pi];
q[ξ_] := Rz[Pi / 2].{{Cos[ξ + Pi] - e}, {sp Sin[ξ + Pi]}, {0}};
```

これを空間固定系XYZで見た座標は，qf(ξ)=RZ(-Φ)RX(-Θ)q(ξ).

```
qf[ξ_, Φ_] := Rz[-Φ].Rx[-Θ].q[ξ];
```

また，地球固定系xyzで見た座標は，qe(ξ)=RZ(t)RZ(-Φ)RX(-Θ)q(ξ).

```
qe[ξ_, Φ_] := Rz[ti[ξ] + Φ].Rz[-Φ].Rx[-Θ].q[ξ];
Print[Simplify[MatrixForm[qe[ξ, Φ]]]];
```

離心率 (0～1)

```
e = 0.1;
```

t=$2\pi/3$となるξを数値的に求める.

```
Plot[Table[ti[ξ], {e, 0, 1, 0.1}], {ξ, 0, 2 Pi}]
sol = FindRoot[2 Pi / 3 == ti[x], {x, 0}]
ξ0 = x /. sol
```

地球半径6356.752 km, 静止衛星高度約35786kmなので，静止衛星高度を1とした地球の半径rは,

```
r = 6356.752 / (35 786 + 6356.752);
```

以下の図で，球は地球，球をつらぬく直線は自転軸，赤，青，緑の円は空間固定系から眺めた3衛星の軌道．なお衛星の楕円軌道の軸と地球の自転軸のなす角Θは,

```
Θ = 45 Pi / 180;
```

これは，衛星が北緯35度（日本付近）で折り返す角度である.

```
s = 0.06; (* 衛星の大きさ *)
pr = 1.4;
Gs = Graphics3D[Sphere[{0, 0, 0}, r]];
Gf = ParametricPlot3D[qf[t, 0], {t, 0, 2 Pi}, PlotStyle → {Red}];
Ge = ParametricPlot3D[qe[t, 0], {t, 0, 2 Pi}];
Gf2 = ParametricPlot3D[qf[t, 2 Pi / 3], {t, 0, 2 Pi}, PlotStyle → {Blue}];
Ge2 = ParametricPlot3D[qe[t, 2 Pi / 3], {t, 0, 2 Pi}];
Gf3 = ParametricPlot3D[qf[t, -2 Pi / 3], {t, 0, 2 Pi}, PlotStyle → {Green}];
Ge3 = ParametricPlot3D[qe[t, -2 Pi / 3], {t, 0, 2 Pi}];
Gz = ParametricPlot3D[{0, 0, z}, {z, -pr, pr}];
Gxz = Graphics3D[{EdgeForm[], FaceForm[Yellow], Opacity[.3],
    Polygon[{{-pr, -pr, 0}, {-pr, pr, 0}, {pr, pr, 0}, {pr, -pr, 0}}]}];
Gr = ParametricPlot3D[(r * 1.01) {Cos[t], Sin[t], 0}, {t, 0, 2 Pi}];
Gd = Graphics3D[{EdgeForm[], FaceForm[Yellow],
    Opacity[.3], Cylinder[{{0, 0, -0.0001}, {0, 0, 0.0001}}, pr]}];
Gsd = Graphics3D[{EdgeForm[], Yellow, Opacity[.3],
    Polygon[{{-pr, -pr, 0}, {pr, -pr, 0}, {pr, pr, 0}, {-pr, pr, 0}}]}];

n = 120; (* ムービーのこま数 *)
dt = 2 Pi / n;
ξ = 0;
ξ2 = 0;
ξ3 = 0;
mv = Table[
  Show[
   Gf,
   Ge = ParametricPlot3D[Rz[-tt].qe[t, 0], {t, 0, 2 Pi}],
   Gif = Graphics3D[{Glow[Red], Sphere[Transpose[qf[ξ, 0]], s]}],
   Gf2,
   Ge2 = ParametricPlot3D[Rz[-tt].qe[t, 2 Pi / 3], {t, 0, 2 Pi}],
   Gif2 = Graphics3D[{Glow[Blue], Sphere[Transpose[qf[ξ2 - ξ0, 2 Pi / 3]], s]}],
   Gf3,
   Ge3 = ParametricPlot3D[Rz[-tt].qe[t, -2 Pi / 3], {t, 0, 2 Pi}],
   Gif3 = Graphics3D[{Glow[Green], Sphere[Transpose[qf[ξ3 + ξ0, -2 Pi / 3]], s]}],
   Gz, Gs, Gr,
   Gp = ParametricPlot3D[(r * 1.01) Rz[-tt].{0, Cos[t], Sin[t]}, {t, 0, 2 Pi}],
   Gsd,
   ξ = ξ + dt / dtpdξ[ξ];
   ξ2 = ξ2 + dt / dtpdξ[ξ2 - ξ0];
   ξ3 = ξ3 + dt / dtpdξ[ξ3 + ξ0];
   PlotRange → {{-pr, pr}, {-pr, pr}, {-pr, pr}}],
  {tt, 0, 2 Pi - 0.0001, dt}
 ]
Export["qzss-0x-120.gif", mv, "gif"]
```